

## CO557 Assignment 4

Version 2021.12.12.15.00

Please ensure that you are always working from the latest version of the assessment, which you can find on the Moodle page in the Assessments section.

### Introduction

This assignment is designed to provide you with a practical understanding of the task of writing the syntax analysis parts of a compiler for a high-level language called Boaz. Boaz is a language designed by me for this particular assessment so you should not expect to find any additional information about it elsewhere on the Web/Internet. This document is the definitive definition/description of Boaz.

**Date set:** 6<sup>th</sup> December 2021.

**Date due:** 24<sup>th</sup> January 2022. ~~21<sup>st</sup> January 2022.~~ Deadline 23:55.

**Submission:** Submit a single zip file of the source files of your complete implementation and the scripts required to compile and/or run it.

### Requirements

The main requirement of this assessment is to write a lexical analyser and recursive-descent parser for the language defined in this document. Boaz source files are stored in a text file whose name ends in a ' `.boaz` ' suffix. Your program must analyse a single Boaz source file named as its only command-line argument. Your program will only be tested with files with the correct suffix.

The grammar to which valid programs must conform is given below. There is a further challenge element, concerned with type checking, worth 20% of the marks.

The program must not output anything other than a single diagnostic message on standard output at the end of the parse. The message must be:

`ok`

for a successful parse or

`error`

for an unsuccessful parse.

A single error in the source file should cause the parser to output the error notification and stop the parse. No error recovery is required. Note that the message must be written to standard output (`System.out` in Java) regardless of whether the parse is successful or not.

You may implement the program in a language of your choice under the constraints that the language must be available to the marker on raptor and it must be possible for the marker both to compile (where required by the language) and execute your code without having to install a particular IDE or any other software, such as libraries or build tools. These constraints are important. Any libraries required must either be bundled with your submission or already pre-installed and accessible to the marker.

## Grammar for Boaz

Here is the grammar of the Boaz language. Note that this has been expressed in a slightly different form from that in the book/slides. You should use this grammar as the basis for structuring your parser.

Each rule of the grammar has a 'non-terminal' symbol on the left-hand side of the '::=' symbol. On the right-hand side of '::=' is a mix of 'terminal' symbols, non-terminal symbols and grammatical notation, terminated with a semicolon. The meaning of the grammatical notation is described below.

- The notation `: : =` means 'is defined as'.
- A semicolon marks the end of each non-terminal 'rule'.
- Parentheses group terminal and non-terminal symbols as a single item.
- A `?` means the single preceding item is optional or the preceding parenthesized items as a group are optional. For instance:  
`a b ? c` means that both `ac` and `abc` are valid and  
`(a b) ? c` means that both `c` and `abc` are valid.
- One or more items enclosed between curly brackets means the enclosed items occur zero or more times. For instance:  
`{ a b }` means that `ab` may occur 0, 1, 2 or more times; e.g., `abab`, `abababab`, `ababababababab` are all valid.
- A `|` separates alternatives. For instance:  
`{ a | b | c }` means that either `a` or `b` or `c` is valid.
- Items all in upper-case are terminal symbols: an identifier (`IDENTIFIER`), keyword (`BEGIN`, `IF`, etc.), integer constant (`INT_CONST`) or character constant (`CHAR_CONST`).
- Items enclosed in single quotes are terminal symbols, e.g. `' ; '` and `' : = '`.

In the grammar be careful to distinguish between those characters not enclosed between single-quote characters (e.g., `: : =` and `;` ) and those that look similar but are enclosed (e.g., `' : = '` and `' ; '`).

```
program ::=      PROGRAM IDENTIFIER varDecs
                BEGIN statements END ;

varDecs ::=      { varDec ' ; ' } ;

varDec  ::=      type varList ;

type    ::=      INT | CHAR;
```

```

varList ::=      IDENTIFIER { ',' IDENTIFIER } ;

statements ::= { statement ';' } ;

statement ::=  assignStatement |
               ifStatement |
               printStatement |
               whileStatement ;

assignStatement ::= IDENTIFIER ':=' expression ;

ifStatement ::=      IF expression THEN statements
                    ( ELSE statements ) ? FI ;

printStatement ::=  PRINT expression ;

whileStatement ::=  WHILE expression DO statements OD;

expression ::= term { binaryOp term } ;

binaryOp ::=  arithmeticOp | booleanOp | relationalOp ;

arithmeticOp ::=  '+' | '-' | '*' | '/' ;

booleanOp ::=      '&' | '|' ;

relationalOp ::=  '=' | '!=' | '<' | '>' | '<=' | '>=' ;

term ::=          INT_CONST |
                  CHAR_CONST |
                  IDENTIFIER |
                  '(' expression ')' |
                  unaryOp term ;

unaryOp ::=  '-' | '!' ;

```

### Comments

Comments are not permitted in Boaz programs.

### Example program

The following (nonsense) program illustrates the main syntactic features defined by the grammar.

```

program example
int num, sum;
char ch;
begin
    sum := 0;
    num := 1;
    ch := "a";

```

```

while ch < "z"
do
    if num / 3 != 0 & ch != "y"
    then
        sum := sum + num * 2;
        ch := ch + 1;
    else
        sum := sum - num;
        ch := "m";
    fi;
    num := num + 1;
od;
print sum;
print ch;
end

```

### Definitions of terminal symbols: lexical analysis

- The following are all keywords of the language: `begin`, `char`, `do`, `else`, `end`, `fi`, `if`, `int`, `od`, `print`, `program`, `then`, `while`. They are case-sensitive and must be all lower-case in the source. These are represented in the grammar by upper-case versions.
- All symbols within a pair of single quote characters in the grammar are one-character or two-character literals that will be found in the source code; e.g., `'!='`.
- IDENTIFIER: A sequence of one or more alphabetic, digit or underscore characters, of which the first must be alphabetic. E.g., `Ng1_f3` is valid but `1e4` is not. It is not permitted to define a variable that is the same as a keyword.
- INT\_CONST: A sequence of one or more digits (0-9).
- CHAR\_CONST: A single character enclosed within a pair of *double-quote* characters. E.g., `"!"`, `"a"`, etc. Multi-character character constants are not permitted.

### Limited type checking – challenge element (20%)

Type checking involves ensuring that variables have been defined and that types are consistent in assignments and expressions. Identifiers and associated type information are normally stored in a 'symbol table'. So, in order to complete this part, you will need to maintain a symbol table for identifiers.

Some type checking of expressions and assignments is required, so you will also need to associate a 'type' (*boolean*, *char*, *int*) with each expression, according to the following rules:

- If the expression contains at least one `relationalOp`, `booleanOp` or unary boolean negation operator (`!`) then its type is *boolean*.
- If the expression is not *boolean* and it contains at least one `CHAR_CONST` or an IDENTIFIER of type `CHAR` then its type is *char*.
- If the expression is neither *boolean* nor *char* then its type is *int*.

Any of the following should result in a failed parse, with the error message given above being printed and the parse stopping.

- Use of an `IDENTIFIER` either on the left-hand side of an `assignStatement` or in a `term` that has not been declared within a `varDec`.
- Assignment of an expression of one type to a variable of a different type in an `assignStatement`. It is only legal to assign a `char` expression to a variable of type `char`, and an `int` expression to a variable of type `int`. It is not permitted to assign a `boolean` expression to any variable.
- Use of an expression of type `int` or `char` in the expression (condition) of either an `ifStatement` or `whileStatement`. All condition expressions must be of type `boolean`.

### Suggested order of implementation

1. Start by writing a lexical analyser/tokenizer to handle all of the symbols of the Boaz language. Print out details of each token as it is recognised and test with small programs to start and then gradually build up to the example given above. Don't start on the parser until the tokenizer is working.
2. Start writing the parser to parse a `program` with no variables or statements. In other words, just the header and the `begin` and `end` keywords. That requires just four tokens from the tokenizer: `PROGRAM`, `IDENTIFIER`, `BEGIN` and `END`. That will give you confidence that the interaction between parser and tokenizer is working correctly.
3. Parse `varDecs`. As part of this process, start to build a symbol table that records each `IDENTIFIER` along with its type. This will be required later for the type-checking parts.
4. Add parsing of `term` and `expression`. You can test the parsing of these alongside parsing `printStatement`.
5. Add parsing of `whileStatement` but don't worry yet about checking the type of the condition.
6. Add `ifStatement` and `assignStatement`. Again, don't worry about the type checking elements.
7. For the challenge part, you will need to create a simple 'symbol table' that holds information on variable identifiers. Identifiers will be added as each `varDec` is parsed. You will also need to associate a type with each `expression` and `term` as it is parsed. You could either build an explicit 'parse tree' for an expression, in which each node of the tree stores the type of the tree below it, and the overall type of the expression will be at its root; or you could simply have the methods that parse

expressions and terms return the type of that particular element and work out how to combine elements of different types according to the rules given above.

### Parser testing

As you are developing the parser it will be best to test with small examples that gradually add code relating to the most recent feature you have added. For instance, you might start with a program with no variable declarations:

```
program Example begin end
```

and check that it can be parsed successfully. Then systematically remove just one token from that code and check that each variation results in a parsing error.

If you then add parsing for a single variable, you might test with the following code:

```
program Example
int x;
begin
end
```

If that parses ok then you can switch `int` for `char` and check that, and then perform error checking along the lines already described.

Next try declaring multiple variables in the same `varDec`.

And so on ...